

A Formalization of Kakutani’s Dichotomy for Infinite Product Measures

Daniel Rodriguez ✉

Independent researcher

Abstract

Kakutani’s dichotomy (1948) states that an infinite product of probability measures is either absolutely continuous or mutually singular with respect to a second such product built from pairwise locally absolutely continuous factors, according to whether the infinite product of the Hellinger affinities of the factor pairs is positive or zero. We present a formalization of both directions of the theorem in Lean 4 over Mathlib, stated for families indexed by an arbitrary type; to the authors’ knowledge this is the first formalization of the theorem in any proof assistant. The development defines the Hellinger affinity as a total $\mathbb{R}_{\geq 0}^\infty$ -valued function of a pair of measures, proves the singular direction by a cylinder squeeze and the Borel–Cantelli lemma, and proves the absolutely continuous direction by a quantitative comparison of the product measure with a finite-window density measure over the measure-dense algebra of cylinders. Martingale convergence, the Kolmogorov zero–one law, and L^p -completeness arguments are not used. As an application, a family of product Poisson measures on an infinite-dimensional torus indexed by the primes is shown to be mutually singular with product Haar measure exactly for parameters $\sigma \leq 1/2$ and equivalent to it exactly for $\sigma > 1/2$. The development carries a zero-sorry continuous-integration policy, and the main theorems depend only on the axioms `propext`, `Classical.choice`, and `Quot.sound`. Upstreaming to Mathlib is planned as a sequence of six pull requests.

2012 ACM Subject Classification Theory of computation → Logic and verification; Mathematics of computing → Probability and statistics

Keywords and phrases Kakutani dichotomy, infinite product measures, Hellinger affinity, measure theory, Lean, Mathlib

Digital Object Identifier 10.4230/LIPIcs.TBD.2026.0

Supplementary Material *Software (Source Code)*: <https://github.com/idolum-ai/riemann-venue>

Acknowledgements This formalization was developed in the `riemann-venue` repository by the author in collaboration with Claude Fable 5 (Anthropic), which co-developed the Lean proofs, computational experiments, and research notes in interactive sessions under the author’s direction and scope decisions throughout. GPT 5.5 Pro (OpenAI) provided feedback on the accompanying essay and on the repository’s initial outline. All mathematical content is machine-checked or explicitly labeled otherwise; correctness rests on the Lean kernel and the reproducibility scripts, not on trust in any author, human or artificial.

1 Introduction

Let $(\Omega_i)_{i \in I}$ be measurable spaces carrying two families of probability measures μ_i and ν_i with $\mu_i \ll \nu_i$ for every i , and let $\mu = \bigotimes_i \mu_i$ and $\nu = \bigotimes_i \nu_i$ be the product measures on $\prod_i \Omega_i$. Kakutani’s theorem [11] says that the relationship between μ and ν is decided by a single numerical invariant, the infinite product of the Hellinger affinities $H(\mu_i, \nu_i) = \int \sqrt{\frac{d\mu_i}{d\nu_i}} d\nu_i$: if $\prod_i H(\mu_i, \nu_i) > 0$ then $\mu \ll \nu$, and if the product is 0 then μ and ν are mutually singular. When each pair is mutually absolutely continuous, the conclusion sharpens to a dichotomy between equivalence and mutual singularity, with no intermediate regime.

The theorem is a standard instrument of measure-theoretic probability. It is the ancestor of the Feldman–Hájek dichotomy for Gaussian measures [5, 6], it governs when two stochastic



© Daniel Rodriguez;
licensed under Creative Commons License CC-BY 4.0
(Venue TBD; prepared in LIPIcs style for an ITP/CPP-type submission).
Editors: (Editors TBD); Article No. 0; pp. 0:1–0:11



Leibniz International Proceedings in Informatics
LIPIcs Schloss Dagstuhl – Leibniz-Zentrum für Informatik, Dagstuhl Publishing, Germany

models with independent coordinates can be distinguished with certainty from a single sample, and its criterion is effective: since $0 \leq H(\mu_i, \nu_i) \leq 1$, the product is positive precisely when the deficits $1 - H(\mu_i, \nu_i)$ are summable, so verifying either branch of the dichotomy reduces to a convergence estimate.

To our knowledge the theorem had no formalization in any proof assistant before this work; Section 6 and Section 7 record the basis for this claim and the protocol for re-verifying it prior to publication. One structural reason is that the statement needs infinite products of probability measures over arbitrary index sets as first-class objects, and Lean's Mathlib [13, 3] acquired these only recently (`Measure.infinitePi`, built through the Ionescu–Tulcea theorem [10]). A second reason is that the textbook proofs [14] run through martingale convergence and L^2 limit arguments, which are formalizable but drag in machinery whose bookkeeping dominates the mathematics. The routes chosen here avoid that machinery entirely.

The contributions:

- Machine-checked statements and proofs of both directions of Kakutani's dichotomy in Lean 4 over Mathlib, for factor families indexed by an *arbitrary* type. This is stronger than the classical countable statement, and it fell out of the proof routes rather than being a goal.
- A reusable definition of the Hellinger affinity as a total, $\mathbb{R}_{\geq 0}^\infty$ -valued, symmetric function of a pair of measures, with its basic API, including $H(\mu, \nu) = 0 \leftrightarrow \mu \perp \nu$ (Subsection 3.2).
- Supporting lemmas absent from Mathlib at the pinned revision: finite tensorization of densities (`Measure.pi_withDensity`), a product Fubini theorem for the lower Lebesgue integral, positivity and summability bridges for infinite products in $\mathbb{R}_{\geq 0}^\infty$, and window-factorization lemmas over `Measure.infinitePi` (Section 3).
- Proof routes suited to formalization: the singular direction by a cylinder squeeze and Borel–Cantelli, the absolutely continuous direction by a quantitative cylinder comparison. Neither uses martingales, the zero–one law, or L^p completeness (Section 4).
- An application: the family of product Poisson measures over the primes at ratios $p^{-\sigma}$ changes type against product Haar measure exactly at $\sigma = 1/2$ (Section 5).
- A public artifact with zero-sorry CI, axiom audits, and a staged plan for upstreaming the general-purpose layer to Mathlib (Section 7).

The formalization was built as an instrument for a research program on the spectral geometry of the greatest-common-divisor kernel [12]; Section 5 states the one theorem of that program that this paper reports.

2 Mathematical background

2.1 The Hellinger affinity

For probability measures μ, ν on a measurable space Ω and any σ -finite measure λ dominating both, the *Hellinger affinity* (also the Bhattacharyya coefficient [7, 2]) is

$$H(\mu, \nu) = \int_{\Omega} \sqrt{\frac{d\mu}{d\lambda} \cdot \frac{d\nu}{d\lambda}} d\lambda,$$

independent of the choice of λ . It is symmetric, and by Cauchy–Schwarz $0 \leq H(\mu, \nu) \leq 1$, with $H(\mu, \mu) = 1$. The two endpoints carry the geometry: $H(\mu, \nu) = 1$ iff $\mu = \nu$, and $H(\mu, \nu) = 0$ iff $\mu \perp \nu$. The affinity tensorizes over finite products, $H(\mu_1 \otimes \mu_2, \nu_1 \otimes \nu_2) = H(\mu_1, \nu_1) \cdot H(\mu_2, \nu_2)$, which is what makes it the right invariant for product measures.

2.2 The dichotomy

► **Theorem 1** (Kakutani [11]). *For each i in a countable index set let μ_i, ν_i be probability measures on Ω_i with $\mu_i \ll \nu_i$, and set $\mu = \bigotimes_i \mu_i$, $\nu = \bigotimes_i \nu_i$.*

1. *If $\prod_i H(\mu_i, \nu_i) > 0$, then $\mu \ll \nu$.*

2. *If $\prod_i H(\mu_i, \nu_i) = 0$, then $\mu \perp \nu$.*

If moreover $\nu_i \ll \mu_i$ for every i , the first case gives equivalence of μ and ν .

Since every factor lies in $[0, 1]$, the partial products are antitone and the infinite product converges to a limit in $[0, 1]$; positivity of the limit is equivalent to summability of the deficits:

$$\prod_i H(\mu_i, \nu_i) > 0 \iff \sum_i (1 - H(\mu_i, \nu_i)) < \infty \text{ and } H(\mu_i, \nu_i) > 0 \text{ for all } i.$$

The classical proof [11, 14] considers the likelihood ratios $Z_n = \prod_{i \leq n} \frac{d\mu_i}{d\nu_i}$, a nonnegative ν -martingale. In the convergent case the square roots $\sqrt{Z_n}$ form a Cauchy sequence in $L^2(\nu)$, the limit identifies $\frac{d\mu}{d\nu}$, and absolute continuity follows; in the divergent case $Z_n \rightarrow 0$ almost everywhere under ν while μ concentrates where Z_n survives, and a zero-one argument upgrades this to mutual singularity. The formalized proofs reach the same conclusions along different routes (Section 4).

3 The formalization


The development comprises six Lean files plus two application files under `RiemannVenue/Kakutani/` in the artifact [12], built against a pinned Mathlib revision (Section 7). Everything in this section and the next lives in the `MeasureTheory` namespace, most of it under `Measure`; the application (Section 5) lives in the repository's own namespace. Figure 1 shows the milestone ladder; each box was designed, then implemented as one complete file with no placeholders.

3.1 Infinite product measures in Mathlib

Mathlib's `Measure.infinitePi` takes an arbitrary index type ι and a family $\mu : (i : \iota) \rightarrow \text{Measure}(X\ i)$ of probability measures, and produces the product probability measure on $\prod_i X\ i$. The $\mathbb{N}$ -indexed core is the projective limit supplied by the Ionescu–Tulcea theorem [10]; the general case is a Carathéodory extension of the induced content on measurable cylinders. The API this development consumes is the cylinder evaluation (`infinitePi_cylinder`), the finite-marginal identification (`infinitePi_map_restrict`), and the restriction rule for lower Lebesgue integrals of finite-window functions (`lintegral_restrict_infinitePi`). The recency of this layer is the main reason a 1948 theorem was still open to formalization.

3.2 The Hellinger affinity as formalized

```
-- The Hellinger affinity (Bhattacharyya coefficient) of two measures:
-- ∫ √(dμ/dσ · dν/dσ) dσ computed against the canonical dominating measure
-- σ = μ + ν. Total and symmetric; no absolute-continuity hypothesis. -/
noncomputable def hellingerAffinity (μ ν : Measure Ω) : ℝ ≥ 0 ∞ :=
  ∫- x, (μ.rnDeriv (μ + ν) x * ν.rnDeriv (μ + ν) x) ^ (1 / 2 : ℝ) ∂(μ + ν)
```



■ **Figure 1** The milestone ladder. M1–M3 are infrastructure, M4 and M5 are the two directions of the dichotomy (independent of one another), M6 is the application. Arrows are import dependencies.

Three design decisions are packed into this definition. First, the dominating measure is fixed to $\mu + \nu$, so the affinity is total: it needs no absolute-continuity or integrability hypotheses, and an invariance lemma (`hellingerAffinity_eq_integral_rnDeriv_mul_rnDeriv`) recovers the value against any dominating σ -finite measure. Second, the value type is $\mathbb{R}_{\geq 0}^\infty$ and the integral is the lower Lebesgue integral, so no side conditions about integrability arise anywhere in the development; Mathlib’s Kullback–Leibler divergence `kldiv` made the same choice. Third, totality is what makes the singular direction of the dichotomy free of case splits: the characterization

```

136
137 theorem hellingerAffinity_eq_zero_iff [IsFiniteMeasure  $\mu$ ] [IsFiniteMeasure  $\nu$ ] :
138   hellingerAffinity  $\mu$   $\nu$  = 0  $\leftrightarrow$   $\mu \perp_m \nu$ 
139

```

holds with no hypotheses beyond finiteness, so degenerate factor pairs need no separate treatment. The remaining API proves symmetry, $H(\mu, \mu) = 1$, $H(\mu, \nu) \leq 1$ (Hölder with exponents 2, 2), the computation rules under $\mu \ll \nu$ and for `withDensity` reweightings, and tensorization over finite products (`hellingerAffinity_pi`).

3.3 The hypothesis carrier: infima of partial products

The dichotomy’s hypothesis is the value of an infinite product. Mathlib’s `tprod` is unsuitable as a carrier for it: like `tsum`, it returns the junk value 1 for non-multipliable families, and “the infinite product is 0” is exactly the case where multipliability fails. Since every factor is ≤ 1 , the finite partial products form an antitone net, and its infimum is the honest value

of the infinite product. The formalized statements therefore quantify over finite subsets:

$$\prod_{s:\text{Finset } \iota} \prod_{i \in s} H(\mu_i, \nu_i) = 0 \quad \text{versus} \quad 0 < \prod_s \prod_{i \in s} H(\mu_i, \nu_i).$$

A bridge lemma converts between this form and the summability criterion:

```

151 theorem iInf_finsetProd_pos_iff_summable_one_sub {h :  $\kappa \rightarrow \mathbb{R}_{\geq 0 \infty}$ }
152   (h1 :  $\forall i, h\ i \leq 1$ ) :
153   (0 <  $\prod s : \text{Finset } \kappa, \prod i \in s, h\ i$ )
154    $\leftrightarrow (\forall i, h\ i \neq 0) \wedge \text{Summable fun } i \Rightarrow 1 - (h\ i).\text{toReal}$ 
155
156
157

```

Its proof rests on two elementary inequalities, formalized over the reals: the Weierstrass bound $1 - \prod_{i \in s} h_i \leq \sum_{i \in s} (1 - h_i)$ and the exponential bound $\prod_{i \in s} h_i \leq \exp(-\sum_{i \in s} (1 - h_i))$.

3.4 The main statements

The following are quoted verbatim from the artifact. The ambient context is

```

161 variable { $\iota : \text{Type}^*$ } {X :  $\iota \rightarrow \text{Type}^*$ } {mX :  $\forall i, \text{MeasurableSpace } (X\ i)$ }
162   ( $\mu\ \nu : (i : \iota) \rightarrow \text{Measure } (X\ i)$ )
163   [ $\forall i, \text{IsProbabilityMeasure } (\mu\ i)$ ] [ $\forall i, \text{IsProbabilityMeasure } (\nu\ i)$ ]
164
165
166

```

with no constraint on ι . The singular direction (`Singular.lean`):

```

167 theorem infinitePi_mutuallySingular (hac :  $\forall i, \mu\ i \ll \nu\ i$ )
168   (h :  $\prod s : \text{Finset } \iota, \prod i \in s, \text{hellingerAffinity } (\mu\ i) (\nu\ i) = 0$ ) :
169   Measure.infinitePi  $\mu \perp_m \text{Measure.infinitePi } \nu$ 
170
171
172

```


The absolutely continuous direction, in its summability and infimum forms, and the two-sided corollary under mutual local absolute continuity (`Equivalence.lean`):

```

173 theorem infinitePi_absolutelyContinuous_of_summable (hac :  $\forall i, \mu\ i \ll \nu\ i$ )
174   (hsum : Summable fun i => 1 - (hellingerAffinity  $(\mu\ i) (\nu\ i)$ ).toReal) :
175   Measure.infinitePi  $\mu \ll \text{Measure.infinitePi } \nu$ 
176
177
178
179
180 theorem infinitePi_absolutelyContinuous (hac :  $\forall i, \mu\ i \ll \nu\ i$ )
181   (h : 0 <  $\prod s : \text{Finset } \iota, \prod i \in s, \text{hellingerAffinity } (\mu\ i) (\nu\ i)$ ) :
182   Measure.infinitePi  $\mu \ll \text{Measure.infinitePi } \nu$ 
183
184
185 theorem infinitePi_absolutelyContinuous_and_symm
186   (hac :  $\forall i, \mu\ i \ll \nu\ i$ ) (hac' :  $\forall i, \nu\ i \ll \mu\ i$ )
187   (h : 0 <  $\prod s : \text{Finset } \iota, \prod i \in s, \text{hellingerAffinity } (\mu\ i) (\nu\ i)$ ) :
188   Measure.infinitePi  $\mu \ll \text{Measure.infinitePi } \nu \wedge$ 
189   Measure.infinitePi  $\nu \ll \text{Measure.infinitePi } \mu$ 

```

Two remarks on the hypotheses, in the interest of precision. The index type is arbitrary in both directions; Section 4 explains why countability is never needed. And the singular direction carries the local absolute-continuity hypothesis `hac` even though the affinity was defined so as not to need it: the hypothesis enters only because the finite window densities are built from `rnDeriv` against ν_i . The singular parts of the μ_i can only help singularity, so the hypothesis is removable; the generalization is recorded in the file's docstring and deferred to the Mathlib port (Section 7). For the same reason, the packaged disjunction (“absolutely continuous or mutually singular”) is not a separate declaration in the artifact; under `hac` it is a two-line order split on the infimum, and it is scheduled to be added, in that packaged form, during the port. A degenerate-case lemma is included: if any single factor pair is mutually



■ **Figure 2** The two proof routes. Here $H_i = \text{hellingerAffinity}(\mu_i)(\nu_i)$, Z_s is the finite-window likelihood ratio, W the ratio over a window disjoint from t , and μ^∞, ν^∞ the infinite product measures.

200 singular, the products are (`infinitePi_mutuallySingular_of_mutuallySingular`), with
 201 no absolute-continuity hypothesis at all.

202 4 Proof engineering

203 Both routes are summarized in Figure 2. The design constraint common to both: keep every
 204 estimate in $\mathbb{R}_{\geq 0}^\infty$ as long as possible, because each crossing between $\mathbb{R}_{\geq 0}^\infty$ and $\mathbb{R}$ costs finiteness
 205 and nonnegativity side conditions, and those crossings are where formalized analysis stalls.

206 4.1 The singular direction: cylinder squeeze and Borel–Cantelli

207 Suppose the infimum of partial affinity products is 0. For each k extract a finite window
 208 s_k with $\prod_{i \in s_k} H_i < 4^{-k}$, and consider the cylinder event $B_k = \{x \mid Z_{s_k}(x) < 4^{-k}\}$ where
 209 $Z_s(x) = \prod_{i \in s} \frac{d\mu_i}{d\nu_i}(x_i)$. Two one-line estimates squeeze this event from both sides:

- 210 ■ $\mu^\infty(B_k) \leq 4^{-k} \leq 2^{-k}$, because the cylinder density identity (`infinitePi_cylinder_`
 211 `eq_setLIntegral_rnDeriv`, file `Cylinder.lean`) evaluates μ^∞ on a cylinder as the ν^∞ -
 212 integral of Z_{s_k} over it, and the integrand is below 4^{-k} there;
- 213 ■ $\nu^\infty(B_k^c) \leq 2^{-k}$, by Markov's inequality applied to $\sqrt{Z_{s_k}}$ together with the moment
 214 identity $\int \sqrt{Z_s} d\nu^\infty = \prod_{i \in s} H_i$ (`lintegral_prod_rnDeriv_rpow_infinitePi`).

215 Both bounds are geometrically summable, so the Borel–Cantelli lemma applies twice: $S =$
 216 $\limsup_k B_k$ is μ^∞ -null, and its complement, contained in $\limsup_k B_k^c$, is ν^∞ -null. The set
 217 S witnesses mutual singularity directly.

218 This route uses no martingale convergence and no zero–one law, although both exist in
 219 Mathlib and were mapped as fallbacks during design. The payoff is concrete: the entire
 220 proof is elementary integration against cylinders, and it is the reason the statement holds

for arbitrary ι , since only a *sequence* of finite windows is ever extracted from the vanishing infimum.

4.2 The absolutely continuous direction: quantitative comparison

The classical route builds $\lim_n \sqrt{Z_n}$ in $L^2(\nu^\infty)$ and identifies its square as the density of μ^∞ . Formalizing it means moving between $\mathbb{R}_{\geq 0}^\infty$ -valued densities and real-valued L^2 functions repeatedly. The implemented route instead compares measures directly. Its core is a quantitative estimate (`Equivalence.lean`):

```

theorem infinitePi_cylinder_le_withDensity_add [DecidableEq  $\iota$ ]
  (hac :  $\forall i, \mu i \ll \nu i$ ) {t u : Finset  $\iota$ } (htu : t  $\subseteq$  u)
  {S : Set ( $\prod i : u, X i$ )} (hS : MeasurableSet S) :
  Measure.infinitePi  $\mu$  (cylinder u S)
     $\leq$  (Measure.infinitePi  $\nu$ ).withDensity
      (fun x =>  $\prod i \in t, (\mu i).rnDeriv (\nu i) (x i)$ ) (cylinder u S)
      + 2 * (2 - 2 *  $\prod i \in u \setminus t, hellingerAffinity (\mu i) (\nu i)$ ) ^ (1 / 2 :  $\mathbb{R}$ )

```

On any cylinder, μ^∞ exceeds the finite-window density measure $\nu^\infty.\text{withDensity } Z_t$ by at most an error controlled by the affinity product over the coordinates beyond t . The proof is the classical second-moment computation in disguise: with W the likelihood ratio over $u \setminus t$ and $c = \sqrt{W}$, the factorization of window moments over disjoint windows (`lintegral_prod_mul_prod_infinitePi`) gives $\int Z_t (c-1)^2 d\nu^\infty \leq 2 - 2 \prod_{i \in u \setminus t} H_i$, and two applications of Cauchy–Schwarz for the lower Lebesgue integral convert this into the additive bound. Every step is in $\mathbb{R}_{\geq 0}^\infty$; subtraction is truncated, and three small helper lemmas manage the truncation arithmetic.

The extension from cylinders to all measurable sets is a symmetric-difference approximation: measurable cylinders form a set algebra generating the product σ -algebra, and Mathlib’s `MeasureDense` machinery shows every measurable set is approximated in measure by cylinders. A short lemma (`measure_le_add_of_le_add_on_setAlgebra`) transports the bound “ $m_1 \leq m_2 + \beta$ on a generating algebra” to all measurable sets. Given $\nu^\infty(E) = 0$ and $\varepsilon > 0$, summability of the deficits supplies, through `Summable.vanishing`, a finite window t whose tail contributes less than ε ; the density measure vanishes on E because ν^∞ does, and $\mu^\infty(E) \leq \varepsilon$ follows. The only crossing from $\mathbb{R}_{\geq 0}^\infty$ to $\mathbb{R}$ in the whole direction is the Weierstrass bridge inside the tail estimate.

Countability of ι is again never used: a summable deficit family is supported on countably many indices without any need to say so, because the argument only ever selects finite windows. The design documents anticipated a reduction of countable ι to $\mathbb{N}$ by re-indexing; the implemented route made it unnecessary, and the stated theorems are stronger than the designed ones.

4.3 Method: scout first, then implement

The development was written design-first. Before any Lean file was created, a design document fixed the definitions, the statement shapes, the milestone ladder of Figure 1, and, for every milestone, the exact Mathlib lemmas to be consumed, each verified by `grep` against the pinned Mathlib revision and recorded with file and line. The document also carried a risk register with ordered fallbacks. Implementation then proceeded one milestone at a time, each landing as a complete file with no `sorry`; the waves compiled green on the first or near-first attempt.

The register is worth reporting because two of its entries fired. The risk item for the absolutely continuous direction predicted that $\mathbb{R}_{\geq 0}^\infty/\mathbb{R}$ bookkeeping would dominate an L^2

route and prescribed keeping all cylinder identities in $\mathbb{R}_{\geq 0}^{\infty}$ with a single crossing; following that prescription to its end produced the comparison route of Subsection 4.2, and the L^2 machinery was never needed. The risk item for the application predicted friction in a quotient-circle (`AddCircle`) model and named an interval model as fallback; the fallback was taken directly (Section 5). The residue of the one pinned-toolchain workaround that remains (a `set_option` guarding one Fubini proof) is flagged in the upstreaming plan.

5 Application: a phase boundary at the critical exponent

The repository [12] studies the kernel $K(m, n) = \gcd(m, n) / \sqrt{mn}$ and its spectral representations. One of them realizes K through Fourier coefficients of a product Poisson measure on an infinite-dimensional torus indexed by the primes, with per-prime radii $a_p = p^{-\sigma}$. The local model at each prime, kept on $\mathbb{R}$ rather than on a quotient circle, is:

```

280 noncomputable def haarLoc : Measure ℝ :=
281   (ENNReal.ofReal (2 * Real.pi))-1 • volume.restrict (Set.Ioc (-Real.pi) Real.pi)
282
283
284 noncomputable def poissonMeasure (a : ℝ) : Measure ℝ :=
285   haarLoc.withDensity fun θ => ENNReal.ofReal (poissonKernel a θ)
286
```

where $\text{poissonKernel } a \theta = (1 - a^2) / |1 - ae^{i\theta}|^2$ is the classical Poisson kernel, a strictly positive probability density for $0 < a < 1$. A computation (`hellingerAffinity_poissonMeasure`) identifies the abstract affinity of the pair with the repository's analytic benchmark value $\text{hellinger}(a) = (2\pi)^{-1} \int_{-\pi}^{\pi} \sqrt{P_a(\theta)} d\theta$, and an analytic bridge already in the repository (`summable_hellinger_deficit_iff`) shows the deficits $1 - \text{hellinger}(p^{-\sigma})$ are summable over the primes iff $\sigma > 1/2$, with divergence at the endpoint. The mechanism is the classical one: the deficit at small a is comparable to a^2 , so the series behaves like $\sum_p p^{-2\sigma}$, which converges exactly for $\sigma > 1/2$ and diverges at $\sigma = 1/2$ like $\sum_p 1/p$.

Feeding this into the two directions of the dichotomy yields the phase boundary as a single statement (`PoissonDichotomy.lean`):

```

297 theorem poissonProduct_dichotomy {σ : ℝ} (hσ₀ : 0 < σ) :
298   (Measure.infinitePi (fun p : Nat.Primes => poissonMeasure ((p : ℝ) ^ (-σ)))
299     ⊥m Measure.infinitePi (fun _ : Nat.Primes => haarLoc) ↔ σ ≤ 1 / 2) ∧
300   (Measure.infinitePi (fun p : Nat.Primes => poissonMeasure ((p : ℝ) ^ (-σ)))
301     ≪ Measure.infinitePi (fun _ : Nat.Primes => haarLoc) ↔ 1 / 2 < σ)
302
303
```

For $\sigma > 1/2$ the two products are in fact equivalent (`poissonProduct_equivalent`); the backward implications are exclusivity, since a nonzero measure cannot be both absolutely continuous and mutually singular with respect to the same measure. The type change lands exactly at the critical exponent, with the endpoint $\sigma = 1/2$ on the singular side. Within the wider program this theorem is the machine-checked form of a claim the accompanying essay makes about the log-scale boundary behavior of the divisibility geometry; the program's other results and its open problems are catalogued in the repository's status ledger [12] and are out of scope here.

6 Related work

Measure and probability theory have mature formalized foundations in several systems. In Isabelle/HOL, measure theory and the core of probability were developed by Hölzl and Heller [9], and infinite products of probability measures, via projective limits in the Ionescu–Tulcea style, are part of Hölzl's subsequent development [8]; the infrastructure for stating

Kakutani’s theorem has therefore existed in Isabelle for some time. In Coq, MathComp-Analysis provides measure construction and the Lebesgue integral [1]. In Lean, Mathlib [13] contains the Radon–Nikodym theorem, the Lebesgue decomposition, martingale convergence, and, recently, the infinite product measures this work builds on. We searched the Isabelle Archive of Formal Proofs, the Coq ecosystem, and the Lean/Mathlib libraries for a formalization of Kakutani’s theorem and found none; the claim of first formalization is made to the authors’ knowledge, and Section 7 commits to re-verifying it immediately before publication.

On the divergence side, Mathlib contains the Kullback–Leibler divergence, $\mathbb{R}_{\geq 0}^{\infty}$ -valued and standalone. A broader f -divergence framework by Degenne and Luccioli [4] exists as an external Lean development and has begun to be upstreamed; the Hellinger affinity defined here is the f -integral at $f = \sqrt{\cdot}$, and a bridging lemma is planned for whenever the general framework lands (Section 7). The formalization of the Feldman–Hájek dichotomy for Gaussian measures [5, 6], for which the present theorem is a natural stepping stone, remains open in all systems as far as we know.

7 Artifact and upstreaming

Artifact.

The development is part of the public repository `riemann-venue` [12], Apache-2.0, frozen for this article at commit [COMMIT]. The Kakutani layer is:

File (RiemannVenue/Kakutani/)	Contents	Lines
<code>Affinity.lean</code>	Hellinger affinity and API (M1)	197
<code>FiniteProduct.lean</code>	<code>pi_withDensity</code> , lintegral Fubini, tensorization (M2)	143
<code>Cylinder.lean</code>	bridges, cylinder density and moment identities (M3)	274
<code>Singular.lean</code>	singular direction (M4)	208
<code>Equivalence.lean</code>	absolutely continuous direction (M5)	484
<code>PoissonSingular.lean</code>	local Poisson model, singular half of M6	218
<code>PoissonDichotomy.lean</code>	equivalence half and the packaged boundary (M6)	175
<code>SpectralMeasure.lean</code>	Bochner window factorization; spectral application	248

The toolchain is Lean 4 (`leanprover/lean4:v4.32.0-rc1`) with Mathlib pinned at revision `e2361c1`. Continuous integration builds the full repository and enforces a zero-sorry policy by a textual guard in addition to the build. Axiom audits via `#print axioms` on the main declarations (`infinitePi_mutuallySingular`, `infinitePi_absolutelyContinuous_of_summable`, `poissonProduct_dichotomy`) report exactly `propext`, `Classical.choice`, and `Quot.sound`. To build:

```
lake exe cache get && lake build
```

Upstreaming.

The general-purpose layer (everything except the Poisson application) is scheduled for Mathlib as six pull requests, about 31 public declarations in total: (1) the Hellinger affinity file; (2) `pi_withDensity` and the lintegral product Fubini; (3) the infinite-product/summability bridges; (4) the singular direction, with the `hac` hypothesis dropped as discussed in Subsection 3.4; (5) the absolutely continuous direction together with the packaged dichotomy and the equivalence restated as a biconditional; (6) the Bochner window factorization over

infinitePi, which is independent of the rest. PRs 1–3 and 6 have no interdependencies; 4 needs 1–3; 5 needs 4. The plan, including naming deltas, the module-system port, and known drift risks between the pin and current Mathlib, is a document in the repository (notes/mathlib-pr-plan.md).

Verification of the novelty claim.

The claim that no prior formalization exists will be re-checked against the Isabelle AFP, the Coq opam ecosystem, and the Lean/Mathlib and Zulip archives immediately before this article is published in any form, and the hedge “to the authors’ knowledge” will be retained regardless of the outcome.

8 Conclusion

Kakutani’s dichotomy is now a machine-checked theorem, in a form slightly stronger than the classical one, with proof routes chosen so that the formalization consists of elementary integration against cylinders rather than martingale limit theory. The exercise supports a working hypothesis about formalizing graduate-level measure theory: the cost concentrates at value-type crossings, and route selection that minimizes those crossings can matter more than the availability of heavy machinery. Next steps are the removal of the residual absolute-continuity hypothesis on the singular direction, the Mathlib pull-request sequence, and, once a general f -divergence framework is in Mathlib, the bridging lemma tying the affinity to it. The Feldman–Hájek dichotomy is the natural continuation.

References

- 1 Reynald Affeldt and Cyril Cohen. Measure construction by extension in dependent type theory with application to integration. *Journal of Automated Reasoning*, 67(3):28, 2023. doi:10.1007/s10817-023-09671-5.
- 2 Anil Kumar Bhattacharyya. On a measure of divergence between two statistical populations defined by their probability distributions. *Bulletin of the Calcutta Mathematical Society*, 35:99–109, 1943.
- 3 Leonardo de Moura and Sebastian Ullrich. The Lean 4 theorem prover and programming language. In *Automated Deduction – CADE 28*, volume 12699 of *Lecture Notes in Computer Science*, pages 625–635. Springer, 2021. doi:10.1007/978-3-030-79876-5_37.
- 4 Rémy Degenne and Lorenzo Luccioli. TestingLowerBounds: a Lean 4 formalization of f -divergences and lower bounds for hypothesis testing. <https://github.com/RemyDegenne/testing-lower-bounds>, 2024. Accessed July 2026.
- 5 Jacob Feldman. Equivalence and perpendicularity of Gaussian processes. *Pacific Journal of Mathematics*, 8(4):699–708, 1958.
- 6 Jaroslav Hájek. On a property of normal distributions of any stochastic process. *Czechoslovak Mathematical Journal*, 8:610–618, 1958.
- 7 Ernst Hellinger. Neue Begründung der Theorie quadratischer Formen von unendlichvielen Veränderlichen. *Journal für die reine und angewandte Mathematik*, 136:210–271, 1909.
- 8 Johannes Hölzl. *Construction and stochastic applications of measure spaces in higher-order logic*. PhD thesis, Technische Universität München, 2013.
- 9 Johannes Hölzl and Armin Heller. Three chapters of measure theory in Isabelle/HOL. In *Interactive Theorem Proving (ITP 2011)*, volume 6898 of *Lecture Notes in Computer Science*, pages 135–151. Springer, 2011. doi:10.1007/978-3-642-22863-6_12.

- 396 10 Cassius T. Ionescu Tulcea. Mesures dans les espaces produits. *Atti della Accademia Nazionale*
397 *dei Lincei, Rendiconti, Classe di Scienze Fisiche, Matematiche e Naturali* (8), 7:208–211,
398 1949.
- 399 11 Shizuo Kakutani. On equivalence of infinite product measures. *Annals of Mathematics*,
400 49(1):214–224, 1948. doi:10.2307/1969123.
- 401 12 Daniel Rodriguez. riemann-venue: a machine-checked workspace for the finite-place spectral
402 geometry of divisibility. <https://github.com/idolum-ai/riemann-venue>, 2026. Apache-2.0.
403 Artifact frozen at commit [COMMIT] for this article.
- 404 13 The mathlib Community. The Lean mathematical library. In *Proceedings of the 9th ACM*
405 *SIGPLAN International Conference on Certified Programs and Proofs (CPP 2020)*, pages
406 367–381. ACM, 2020. doi:10.1145/3372885.3373824.
- 407 14 David Williams. *Probability with Martingales*. Cambridge University Press, 1991. doi:10.
408 1017/CB09780511813658.